The Torsional Oscillator -- Comparing Linear and Air Drag fits

In[176]:=
```
Clear["Global`*"]; DateString[]
SetDirectory[NotebookDirectory[]];
```
Out[176]=
```
Wed 26 Mar 2025 12:07:07
```

In[178]:=
```
model[V0_, γ_, ω_, ϕ_, Voff_, t_] := V0 Exp[- γ t / 2] Sin[ω t + ϕ] + Voff
```

In[179]:=
```
(* Handy function to get CSV files output by LoggerPro,
stripping off the title headers *)
getfile[filename_] := Select[Import[filename, "CSV"],
  Length[#] == 2 && VectorQ[#, NumberQ] &]
```

In[180]:=
```
damped = getfile["TO-20250220-damped-1.csv"];
airdrag  = getfile["TO-20250221-airdrag-1.csv"];
```

Damped Oscillations

In[182]:=
```
dfit = NonlinearModelFit[damped,
  model[V0, γ, ω, ϕ, Voff, t], {V0, γ, ω, ϕ, Voff}, t]
dfit["ParameterConfidenceIntervalTable"] // Quiet
{ωv, δωv} = dfit["ParameterConfidenceIntervalTableEntries"][[3, {1, 2}]] // Quiet
Show[{ListPlot[damped, PlotStyle → Red],
  Plot[dfit[t], {t, 0, 30}, PlotRange → All]},
 ImageSize → Scaled[0.8], LabelStyle → Larger, Frame → True,
 FrameLabel → {"t (s)", "V(t)"}, PlotRange → All, PlotLabel → "Damped"]
```

Out[182]=
```
FittedModel[ -0.00317 + 1.26 ≪1≫ Sin[1.32 - ≪1≫] ]
```

Out[183]=

	Estimate	Standard Error	Confidence Interval
V0	−1.25665	0.000246297	{−1.25713, −1.25616}
γ	0.141641	0.0000434628	{0.141556, 0.141727}
ω	4.41284	0.0000216336	{4.41279, 4.41288}
ϕ	−1.32032	0.000193248	{−1.3207, −1.31994}
Voff	−0.00316856	0.0000561695	{−0.00327866, −0.00305846}

Out[184]=
```
{4.41284, 0.0000216336}
```

Out[185]=



In[186]:=
```
dresiduals =
  Table[{damped[[i, 1]], damped[[i, 2]] - dfit[damped[[i, 1]]]}, {i, 1, Length[damped]}];
```

In[187]:=
```
ListPlot[{damped, dresiduals}, PlotRange → {{5, 10}, {-0.1, 0.1}},
 ImageSize → Scaled[0.8], LabelStyle → Larger, Frame → True,
 FrameLabel → {"t (s)", "V(t)"}, PlotLabel → "Damped Residuals"]
```

Out[187]=



In[188]:=
```
dampedRMSE = Sqrt[dfit["EstimatedVariance"]]
```

Out[188]=
```
0.00687428
```

Air drag

In[189]:=

```
afit = NonlinearModelFit[airdrag,
  model[V0, γ, ω, ϕ, Voff, t], {V0, γ, ω, ϕ, Voff}, t]
afit["ParameterConfidenceIntervalTable"] // Quiet
{ωa, δωa} = afit["ParameterConfidenceIntervalTableEntries"][[3, {1, 2}]] // Quiet
Show[{ListPlot[airdrag, PlotStyle → Red, PlotLegends → {"Data"}],
  Plot[afit[t], {t, 0, 30}, PlotRange → All, PlotLegends → {"Fit"}]},
 ImageSize → Scaled[0.8], LabelStyle → Larger, Frame → True,
 FrameLabel → {"t (s)", "V(t)"}, PlotRange → All, PlotLabel → "Air Drag"]
```

Out[189]=

```
FittedModel[ −0.0138 + 0.777 ≪1≫ Sin[0.011 + ≪1≫] ]
```

Out[190]=

	Estimate	Standard Error	Confidence Interval
V0	0.777428	0.0013233	{0.774834, 0.780022 }
γ	0.29767	0.000715016	{0.296269, 0.299072 }
ω	2.6973	0.000354199	{2.6966, 2.69799 }
ϕ	0.0109863	0.00168016	{0.00769298, 0.0142796 }
Voff	−0.0138128	0.000218558	{−0.0142412 , −0.0133844 }

Out[191]=

```
{2.6973, 0.000354199}
```

Out[192]=



The fit decays too rapidly, especially at late times. This is because for small velocity, the v^2 damping of air drag is small compared to the v^1 linear damping of the fit.

Comparing models

In[193]:=
```
Sqrt[dfit["EstimatedVariance"]]
```
Out[193]=
```
0.00687428
```

In[194]:=
```
Sqrt[afit["EstimatedVariance"]]
```
Out[194]=
```
0.0266907
```

The larger typical error for the air drag is clear -- another sign that it's just the wrong model.

Compare Residuals

Damped Residuals

In[195]:=
```
dresiduals =
  Table[{damped[[i, 1]], damped[[i, 2]] - dfit[damped[[i, 1]]]}, {i, 1, Length[damped]}];
```

In[196]:=
```
ListPlot[{damped, dresiduals},
 PlotRange → All, PlotLegends → {"Data", "Residuals"},
 ImageSize → Scaled[0.8], LabelStyle → Larger, Frame → True,
 FrameLabel → {"t (s)", "V(t)"}, PlotLabel → "Damped Residuals"]
```
Out[196]=



In[197]:=
```
ListPlot[{damped, dresiduals},
 PlotRange → {{5, 10}, {-0.1, 0.1}}, PlotLegends → {"Data", "Residuals"},
 ImageSize → Scaled[0.8], LabelStyle → Larger, Frame → True,
 FrameLabel → {"t (s)", "V(t)"}, PlotLabel → "Zoomed Damped Residuals"]
```

Out[197]=



The residuals are all quite small. There is a slight trend for larger residuals when the velocity is high (as the oscillator moves through the origin).

Air Drag Residuals

In[198]:=
```
aresiduals = Table[
   {airdrag[[i, 1]], airdrag[[i, 2]] - afit[airdrag[[i, 1]]]}, {i, 1, Length[airdrag]}];
```

In[199]:=
```
ListPlot[{airdrag, aresiduals},
 PlotRange → All, PlotLegends → {"Data", "Residuals"},
 ImageSize → Scaled[0.8], LabelStyle → Larger, Frame → True,
 FrameLabel → {"t (s)", "V(t)"}, PlotLabel → "Air Drag Residuals"]
```

Out[199]=



In[200]:=
```
ListPlot[{airdrag, aresiduals}, PlotRange → {All, {-0.1, 0.1}},
 PlotLegends → {"Data", "Residuals"}, ImageSize → Scaled[0.8],
 LabelStyle → Larger, Frame → True, FrameLabel → {"t (s)" , "V(t)"},
 PlotLabel → "Zoomed Air Drag Residuals"]
```

Out[200]=



A systematic pattern is clear here, and it is generally worse at the peaks, meaning the linear fit systemati cally gets the peak heights wrong.